Lecture 13: Simple Quasi-Static Moment Method Analysis of a Microstrip.

Computational electromagnetics (CEM) can provide accurate solutions for Z_0 and other microstrip properties of interest including plots of $\bar{E}$ and $\bar{H}$ everywhere in space, and $\bar{J}_s$ and ρ_s on the land or the ground plane. This can be accomplished regardless of the cross-sectional geometry of the microstrip, the thickness of the land or its conductivity.

The method of moments (MM) is a very popular CEM technique. It is particularly useful for planar geometries such as microstrip, stripline, conformal antennas, etc.

The MM was popularized by R. F. Harrington in 1965 with his book "Field Computations by Moment Methods." Today, it is one of the most widely used CEM techniques.

We'll illustrate the MM technique with a solution to a quasi-static microstrip immersed in an infinite dielectric as shown:



That is, there is no substrate, *per se*.

© 2012 Keith W. Whites

Integral Equation

We'll imagine that a time harmonic voltage source has been applied across the two conductors:



This causes a charge accumulation as shown.

Next, the image method will be employed to create an equivalent problem for the fields in the upper half space ($y \geq 0$):



In a previous EM course, you've likely learned that the electric potential Φ_e at a point $\overline{r}$ in a homogeneous space produced by a line charge density $\rho_l(\overline{r}')$ is given by

$$\begin{aligned}\Phi_e(\bar{r}) &= \frac{1}{2\pi\varepsilon}\int_{C'}\rho_l(\bar{r}')\ln\left(\frac{1}{|\bar{r}-\bar{r}'|}\right)dl' \\ &= -\frac{1}{2\pi\varepsilon}\int_{C'}\rho_l(\bar{r}')\ln\left(\sqrt{(x-x')^2+(y-y')^2}\right)dx'\end{aligned} \tag{1}$$

(For example, see J. Van Bladel, *Electromagnetic Fields*. New York: Hemisphere Publishing, 1985.)

It is very important to realize that this contour C' must include all charge densities in the space, which means we must include both conductors in this integral.

To develop an equation from which we can solve for the charge density, we'll apply the boundary condition

$$\Phi_e(\bar{r}) = V \qquad \forall\bar{r}\in\{\text{upper strip}\} \tag{2}$$

Now, using (2) in (1) and accounting for both the $+\rho_l$ and $-\rho_l$ strips yields

$$V = -\frac{1}{2\pi\varepsilon}\left\{\int_{\text{top}}\rho_l(\bar{r}')\ln\left[\sqrt{(x-x')^2+\cancelto{0}{(d-d)^2}}\right]dx' - \int_{\text{bottom}}\rho_l(\bar{r}')\ln\left[\sqrt{(x-x')^2+(d+d)^2}\right]dx'\right\}$$

or $$\boxed{V = -\frac{1}{2\pi\varepsilon}\int_0^w\rho_l(\bar{r}')\left[\ln(|x-x'|)-\ln\left(\sqrt{(x-x')^2+4d^2}\right)\right]dx'} \tag{3}$$

Recall that the unknown in (3) is the line charge density ρ_l. But how do we solve for this function? It varies along the strip so we can't simply "pull" it out of the integral.

Actually, (3) is called an integral equation because the unknown function is located in an integrand. You most likely haven't encountered such equations before. Integral equations are very difficult to solve analytically. We'll use a numerical solution method instead.

Basis Function Expansion

In the moment method, we first expand ρ_l in a set of basis functions. For a simple MM solution, here we'll use pulse basis functions and divide the strips into N uniform sections:



where $$\rho_l\left(\vec{r}'\right) \approx \sum_{n=1}^{N} \alpha_n P_n\left(x'; x_{n-1}, x_n\right) \tag{4}$$

What is not known in (4) are the amplitudes α_n of the line charge density expansion. These are just numbers. So, instead of directly solving for the spatial variation of ρ_l in (3), now we'll just be computing these N numbers, α_n. Much simpler!

However, we need to allow enough "degrees of freedom" in this basis function expansion (4) so that an accurate solution can be found. This is accomplished by choosing the proper type of expansion functions, a large enough N, etc.

The next step in the MM solution is to substitute (4) into (3)

$$V = -\frac{1}{2\pi\varepsilon}\int_0^w \left[\sum_{n=1}^{N} \alpha_n P_n\left(x'; x_{n-1}, x_n\right)\right] G\left(\left|x - x'\right|\right) dx' \tag{5}$$

where $$G\left(\left|x - x'\right|\right) = \ln\left(\left|x - x'\right|\right) - \ln\left(\sqrt{\left(x - x'\right)^2 + 4d^2}\right) \tag{6}$$

and is called the Green's function.

We can interchange the order of integration and summation in (5) since these are linear operators, except perhaps when $x = x'$. In this case, the integrand becomes singular. We'll consider this situation later in this lecture.

Then, (5) becomes

$$V = -\frac{1}{2\pi\varepsilon}\sum_{n=1}^{N}\alpha_n \int_0^w P_n\left(x'; x_{n-1}, x_n\right) G\left(\left|x - x'\right|\right) dx'$$

or

$$V = -\frac{1}{2\pi\varepsilon}\sum_{n=1}^{N}\alpha_n \int_{x_{n-1}}^{x_n} G\left(\left|x - x'\right|\right) dx' \tag{7}$$

Testing the Integral Equation

In (7), we have N unknown coefficients α_n to solve for, but only a single equation. We will generate a total of N equations by evaluating (7) at N points along the (top) strip. This process is called "testing" the integral equation. We'll test (7) at the centers of each of the N segments, x_m, giving

$$\boxed{V = -\frac{1}{2\pi\varepsilon}\sum_{n=1}^{N}\alpha_n \int_{x_{n-1}}^{x_n} G\left(\left|x_m - x'\right|\right) dx' \qquad m = 1,\ldots,N} \tag{8}$$

This is the final system of equations that we will use to solve for all the coefficients α_n.

Matrix Equation

It is helpful to cast (8) into the form of a matrix equation

$$\underbrace{[V]}_{N\times 1} = \underbrace{[Z]}_{N\times N} \cdot \underbrace{[\alpha]}_{N\times 1} \tag{9}$$

where

$$V_m = V \tag{10a}$$

$$\alpha_n = \alpha_n \tag{10b}$$

$$Z_{mn} = -\frac{1}{2\pi\varepsilon} \int_{x_{n-1}}^{x_n} G\left(\left|x - x'\right|\right) dx' \qquad (10c)$$

The numerical solution to (9) is accomplished by "filling" or "populating" [V] and [Z], then solving a system of linear, constant coefficient equations. In particular, for

- [V] – choose V = 1 V in (10a), for example.
- [Z] – compute (10c) analytically, if possible, or by numerical integration.

The "filling" of [V] is very simple, while filling [Z] is a bit more difficult. In this quasi-static microstrip example, though, it is possible to evaluate all of the terms analytically since a simple anti-derivative is available.

In particular, with the center of the strip located at the origin as shown:



then the electrostatic potential at point $\bar{r}$ produced by a strip of width Δ supporting a constant line charge density ρ_l is given by

$$\Phi_e\left(\bar{r}\right) = -\frac{\rho_l}{2\pi\varepsilon} \int_{-\Delta/2}^{\Delta/2} \ln\left[\sqrt{\left(x - x'\right)^2 + y^2}\right] dx' \qquad (11)$$

This integral can be evaluated analytically since the integrand, as it turns out, has a relatively simple anti-derivative. Performing the integration in (11), we find that when $\bar{r}$ does not lie anywhere on the strip, the potential is

$$\Phi_e(\bar{r}) = -\frac{\rho_l}{2\pi\varepsilon}\Big\{(x+\Delta/2)\ln\left[(x+\Delta/2)^2+y^2\right] - (x-\Delta/2)\ln\left[(x-\Delta/2)^2+y^2\right] - 2\Delta + 2y\left[\tan^{-1}\left(\frac{x+\Delta/2}{y}\right) - \tan^{-1}\left(\frac{x-\Delta/2}{y}\right)\right]\Big\} \tag{12}$$

while if $x = y = 0$ (at the center of the strip), then

$$\Phi_e(0) = \frac{\rho_l}{2\pi\varepsilon}\left[1-\ln(\Delta/2)\right] \tag{13}$$

Using (12) and (13) in (10c), it can be shown that for $m \neq n$

$$\boxed{Z_{mn} = -\frac{1}{4\pi\varepsilon}\Big\{(\Delta_{mn}+\Delta/2)\ln\left[\frac{(\Delta_{mn}+\Delta/2)^2}{(\Delta_{mn}+\Delta/2)^2+4d^2}\right] - (\Delta_{mn}-\Delta/2)\ln\left[\frac{(\Delta_{mn}-\Delta/2)^2}{(\Delta_{mn}-\Delta/2)^2+4d^2}\right] - 4d\left[\tan^{-1}\left(\frac{\Delta_{mn}+\Delta/2}{2d}\right) - \tan^{-1}\left(\frac{\Delta_{mn}-\Delta/2}{2d}\right)\right]\Big\}} \tag{14}$$

where

$$\boxed{\Delta_{mn} \equiv x_m - x_n} \tag{15}$$

while if $m = n$, then

$$\boxed{\begin{aligned} Z_{mm} &= \frac{\Delta}{2\pi\varepsilon}\left[1-\ln\left(\Delta/2\right)\right] \\ &\quad + \frac{1}{4\pi\varepsilon}\left\{\Delta\ln\left[\left(\Delta/2\right)^2+4d^2\right]-2\Delta+8d\tan^{-1}\left(\frac{\Delta}{4d}\right)\right\} \end{aligned}} \tag{16}$$

Static Moment Method Solution for a Microstrip in an Infinite Dielectric

Keith W.Whites

Last Modified on October 1, 2003

- The microstrip substrate and the infinite space above it are assumed to be the same material having relative permittivity "ϵrbackgnd." Using pulse expansion and point matching for the MM solution with "numcells" segments of width "Δ" uniformly distributed across the infinitely-thin strip, which has a width-over-separation ratio "Woverd." The voltage of the strip conductor is "Vapplied" volts with respect to the ground plane.

- Revisions:
  →10/1/03:First completed.

In[17]:=
```
<< Graphics`Graphics`
```

Enter the geometrical parameters and the applied voltage.

In[18]:=
```
ClearAll[Woverd, ϵrbackgnd, Vapplied, d, width, Δ, c0, ϵ0, Δx, m, n, Zmn, ϵr, Zmatrix,
  Vvector, α, α0, numcells, CPUL, C0PUL, Z0, ϵreff, ϵe, Wod, Z0approx]

Woverd := 5. ;
ϵrbackgnd = 1. ;
Vapplied = 1. ;
```

Compute the moment method solution. "α" and "α0" are the line charge density coefficients when the background relative permittivity is ϵrbackgnd and 1, respectively. Without loss of generality, assume a spacing of d=1. between the microstrip and the ground plane.

In[42]:=
```
d = 1. ;
width := Woverd * d ;
Δ := width / numcells ;

c0 = 2.998 * 10^8 ;
ϵ0 = 8.854 * 10^-12 ;

Δx[m_, n_] := Δ * (m - n)
Zmn[m_, n_, ϵr_] :=
 -1 / (4 * Pi * ϵr * ϵ0) *
   ((Δx[m, n] + Δ / 2) * Log[(Δx[m, n] + Δ / 2)^2 / ((Δx[m, n] + Δ / 2)^2 + 4 * d^2)] -
    (Δx[m, n] - Δ / 2) * Log[(Δx[m, n] - Δ / 2)^2 / ((Δx[m, n] - Δ / 2)^2 + 4 * d^2)] -
    4 * d * (ArcTan[2 * d, Δx[m, n] + Δ / 2] - ArcTan[2 * d, Δx[m, n] - Δ / 2])) /; m ≠ n
Zmn[m_, n_, ϵr_] := Δ / (2 * Pi * ϵr * ϵ0) * (1 - Log[Δ / 2]) +
  1 / (4 * Pi * ϵr * ϵ0) * (Δ * Log[Δ^2 / 4 + 4 * d^2] - 2 * Δ + 8 * d * ArcTan[Δ / (4 * d)])

Zmatrix[ϵr_] := Table[Zmn[m, n, ϵr], {m, numcells}, {n, numcells} ]
Vvector := Table[Vapplied, {numcells}] ;

α := LinearSolve[Zmatrix[ϵrbackgnd], Vvector]
α0 := LinearSolve[Zmatrix[1], Vvector]
```

- **Choose the number of pulse basis functions "numcells"then compute the capacitance per unit length "CPUL" assuming a background relative permittivity "ϵrbackgnd." The characteristic impedance of the microstrip "Z0" and the effective relative permittivity "ϵreff" of a TEM wave propagating on this microstrip are computed from both CPUL as well as "C0PUL," which is the capacitance per unit length of the microstrip with a background relative permittivity equal to 1.**

```
In[23]:= numcells := 50

CPUL := Total[α] * Δ
C0PUL := Total[α0] * Δ

Z0 := (c0 * Sqrt[CPUL * C0PUL])^-1
ϵreff := CPUL / C0PUL

N[Woverd]
N[Z0]
N[ϵreff]

Out[28]= 5.

Out[29]= 49.9727

Out[30]= 1.
```

- **Plot the amplitudes of the line charge density coefficients α. It can be shown theoretically that the line charge density should approach infinity at the edges of the strip. This is called the "edge effect." In the MM solution, we have not allowed that physical characteristic to occur. Nevertheless, the line charge density is becoming large near the edges.**

```
In[31]:= BarChart[α, AxesLabel → {"Basis #", "ρl (C/m)"}, BarSpacing → -0.15,
  PlotLabel → "Microstrip in Infinite Dielectric"]
```




```
Out[31]= - Graphics -
```

- **The variation of C, Z0 and ϵreff are next observed as the number of basis functions is increased. This is called a "convergence study." It is not known a priori how many basis functions are needed in a MM solution to provide an accurate solution. A convergence study should show that the physical quantities are smoothly approaching an asymptote as the number of basis functions increases.**

In[32]:=
```
ListPlot[Table[{numcells, CPUL}, {numcells, 50}], PlotJoined → True,
  AxesLabel → {"# basis", "C [F/m]"}, PlotLabel → "Infinite Dielectric", PlotRange → All,
  PlotStyle → RGBColor[1, 0, 0]]
```



Out[32]= - Graphics -

In[33]:=
```
ListPlot[Table[{numcells, Z0}, {numcells, 50}], PlotJoined → True,
  AxesLabel → {"# basis", "Z0 [Ω]"}, PlotLabel → "Infinite Dielectric", PlotRange → All,
  PlotStyle → RGBColor[1, 0, 0]]
```



Out[33]= - Graphics -

```
In[34]:= ListPlot[Table[{numcells, ∊reff}, {numcells, 50}], PlotJoined → True,
           AxesLabel → {"# basis", "∊reff"}, PlotLabel → "Infinite Dielectric",
           PlotRange → {0, ∊rbackgnd + 1}, PlotStyle → RGBColor[1, 0, 0]]
```




```
Out[34]= - Graphics -
```

- **Lastly, plot and list Z0 from the MM solution as a function of W/d and compare with the approximate solution given in (3.196) of your text. This approximate expression was presumably obtained by researchers curve fitting numerically accurate results, such as those from a MM solution like this one. These two solutions should be in close agreement only for ∊rbackgnd=1.**

- **Approximate formula for the characteristic impedance of a quasi-static microstrip from (3.196) in the text.**

```
In[35]:= ∊e[Wod_, ∊r_] := (∊r + 1) / 2 + (∊r - 1) / 2 * 1 / Sqrt[1 + 12 / Wod]
         Z0approx[Wod_, ∊r_] := 60 / Sqrt[∊e[Wod, ∊r]] * Log[8 / Wod + Wod / 4] /; Wod ≤ 1
         Z0approx[Wod_, ∊r_] := 120 * Pi / (Sqrt[∊e[Wod, ∊r]] * (Wod + 1.393 + 0.667 * Log[Wod + 1.444])) /;
           Wod > 1

In[38]:= numcells := 50
         plot1 := ListPlot[Table[{Woverd, Z0}, {Woverd, 0.1, 10, 0.2}], AxesLabel → {"W/d", "Z0 [Ω]"},
            PlotJoined → True, PlotLabel → "Moment Method/Infinite Dielectric",
            PlotStyle → RGBColor[1, 0, 0]] ;
         plot2 := ListPlot[Table[{Woverd, Z0approx[Woverd, ∊rbackgnd]}, {Woverd, 0.1, 10, 0.2}],
            AxesLabel → {"W/d", "Z0approx [Ω]"}, PlotJoined → True, PlotLabel → "Approximate Formula",
            PlotStyle → RGBColor[0, 0, 1]] ;
         Show[{plot1, plot2}, AxesLabel → {"W/d", "Z0 [Ω]"}, PlotLabel → "Both"]

Out[41]= $Aborted
```

```
In[170]:= numcells := 50
TableForm[Table[{Woverd, Z0, Z0approx[Woverd, ∈rbackgnd]}, {Woverd, 0.2, 10, 0.4}],
 TableHeadings → {None, {"W/d", "Z0 [Ω]", "Z0approx [Ω]"}}]
```

Out[171]//TableForm=

W/d	Z0 [Ω]	Z0approx [Ω]
0.2	221.672	221.408
0.6	156.375	156.087
1.	126.82	126.613
1.4	108.104	108.016
1.8	94.7633	94.7707
2.2	84.6318	84.6128
2.6	76.6168	76.5471
3.	70.0886	69.9704
3.4	64.6522	64.4942
3.8	60.045	59.8562
4.2	56.0844	55.8728
4.6	52.6389	52.411
5.	49.6112	49.372
5.4	46.9275	46.6811
5.8	44.5309	44.2801
6.2	42.3765	42.1237
6.6	40.4284	40.1753
7.	38.6576	38.4057
7.4	37.0406	36.7908
7.8	35.5577	35.3107
8.2	34.1925	33.949
8.6	32.9312	32.6916
9.	31.7624	31.5268
9.4	30.6758	30.4446
9.8	29.6631	29.4363